

clear — Clear memory

[Syntax](#) [Description](#) [Remarks and examples](#) [Also see](#)

Syntax

```
clear
```

```
clear [ mata | results | matrix | programs | ado ]
```

```
clear [ all | * ]
```

Description

`clear`, by itself, removes data and value labels from memory and is equivalent to typing

```
. version 13
. drop _all           (see [D] drop)
. label drop _all     (see [D] label)
```

`clear mata` removes Mata functions and objects from memory and is equivalent to typing

```
. version 13
. mata: mata clear    (see [M-3] mata clear)
```

`clear results` eliminates stored results from memory and is equivalent to typing

```
. version 13
. return clear        (see [P] return)
. ereturn clear       (see [P] return)
. sreturn clear       (see [P] return)
. _return drop _all   (see [P] _return)
```

`clear matrix` eliminates from memory all matrices created by Stata's `matrix` command; it does not eliminate Mata matrices from memory. `clear matrix` is equivalent to typing

```
. version 13
. return clear        (see [P] return)
. ereturn clear       (see [P] return)
. sreturn clear       (see [P] return)
. _return drop _all   (see [P] _return)
. matrix drop _all    (see [P] matrix utility)
. estimates drop _all (see [R] estimates)
```

`clear programs` eliminates all programs from memory and is equivalent to typing

```
. version 13
. program drop _all    (see [P] program)
```

`clear ado` eliminates all automatically loaded ado-file programs from memory (but not programs defined interactively or by do-files). It is equivalent to typing

```
. version 13
. program drop _allado (see [P] program)
```

`clear all` and `clear *` are synonyms. They remove all data, value labels, matrices, scalars, constraints, clusters, stored results, sersets, and Mata functions and objects from memory. They also close all open files and postfiles, clear the class system, close any open Graph windows and dialog boxes, drop all programs from memory, and reset all timers to zero. They are equivalent to typing

```
. version 13
. drop _all           (see [D] drop)
. label drop _all     (see [D] label)
. matrix drop _all    (see [P] matrix utility)
. scalar drop _all    (see [P] scalar)
. constraint drop _all (see [R] constraint)
. cluster drop _all   (see [MV] cluster utility)
. file close _all     (see [P] file)
. postutil clear      (see [P] postfile)
. _return drop _all   (see [P] _return)
. discard             (see [P] discard)
. program drop _all   (see [P] program)
. timer clear         (see [P] timer)
. mata: mata clear    (see [M-3] mata clear)
```

Remarks and examples

[stata.com](http://www.stata.com)

You can clear the entire dataset without affecting macros and programs by typing `clear`. You can also type `clear all`. This command has the same result as `clear` by itself but also clears matrices, scalars, constraints, clusters, stored results, sersets, Mata, the class system, business calendars, and programs; closes all open files and postfiles; closes all open Graph windows and dialog boxes; and resets all timers to zero.

► Example 1

We load the `bpwide` dataset to correct a mistake in the data.

```
. use http://www.stata-press.com/data/r13/bpwide
(fictional blood pressure data)
. list in 1/5
```

	patient	sex	agegrp	bp_bef~e	bp_after
1.	1	Male	30-45	143	153
2.	2	Male	30-45	163	170
3.	3	Male	30-45	153	168
4.	4	Male	30-45	153	142
5.	5	Male	30-45	146	141

```
. replace bp_after = 145 in 3
(1 real change made)
```

We made another mistake. We meant to change the value of `bp_after` in observation 4. It is easiest to begin again.

```
. clear
. use http://www.stata-press.com/data/r13/bpwide
(fictional blood pressure data)
```

Also see

[D] **drop** — Drop variables or observations

[P] **discard** — Drop automatically loaded programs

[U] **11 Language syntax**

[U] **13 Functions and expressions**